
Obrázek 3.8: Část výsledné kuželové plochy

3.4 Řešení Příkladu 1 (str.55) v programu Maple

Zadání: Vyšetřete kvadriku [2], [5]

7x2 + 6y2 + 5z2 − 4xy − 4yz − 22x+ 24y + 2z + 30 = 0. (3.1)

Kompletní kód řešení v Maple:

Poznámka: Výstupy některých příkazů kódu nejsou v textu uvedeny. Většinou
se jedná o případy, kdy výstup pouze kopíruje zadaný příkaz.

> restart;

> with(LinearAlgebra): with(linalg): with(plots):

Obecnou rovnici kvadriky můžeme zapsat a vytvořit užitím matice kvadriky
K:

> X:=Vector[row]([x,y,z,1]);

> K:=Matrix(a,1..4,1..4,shape=symmetric);

K :=

⎡
⎢⎢⎣
a(1, 1) a(1, 2) a(1, 3) a(1, 4)
a(1, 2) a(2, 2) a(2, 3) a(2, 4)
a(1, 3) a(2, 3) a(3, 3) a(3, 4)
a(1, 4) a(2, 4) a(3, 4) a(4, 4)

⎤
⎥⎥⎦

> Kvadrika:=sort(expand(X.K.Transpose(X)),[x,y,z])=0;

Kvadrika := a(1, 1)x2 + 2a(1, 2)x y + 2a(1, 3)x z + a(2, 2) y2 + 2a(2, 3) y z

+a(3, 3) z2 + 2a(1, 4)x+ 2a(2, 4) y + 2a(3, 4) z + a(4, 4) = 0

Rovnice kvadriky dle zadání:

> RovKv:=7*x^2+6*y^2+5*z^2-4*x*y-4*y*z-22*x+24*y+2*z+30=0;

Hodnoty koeficientů rovnice dané kvadriky, potřebné pro vytvoření její matice,
získáme porovnáním obecného tvaru rovnice kvadriky Kvadrika s danou kon-
krétní rovnicí RovKv. To vede na soustavu jednoduchých rovnic SoustRovKoef,
z nichž každá má jako neznámou jeden z koeficientů a(1,1), ..., a(4,4)
(viz následující řádky kódu).

Poznámka:Mohli jsme také určit každý koeficient zvlášť opakovaným použitím
funkce coeff na levou stranu rovnice RovKv.

> SoustRovKoef:={coeffs(lhs(collect(RovKv-Kvadrika,

[x,y,z],distributed)))};

> KoefRovKv:=solve(SoustRovKoef,{a(1,1),a(1,2),a(1,3),

a(1,4),a(2,2),a(3,3),a(2,3),a(2,4),a(3,4),a(4,4)});

Rovnosti a(i,j)=číslo převedeme na přiřazovací příkazy užitím funkce assign.

> assign(KoefRovKv);

Tím se do obecného tvaru matice kvadriky K dosadí konkrétní hodnoty. Matice
K dané kvadriky má pak tvar:

> K; ⎡
⎢⎢⎣

7 −2 0 −11
−2 6 −2 12
0 −2 5 1

−11 12 1 30

⎤
⎥⎥⎦

Diskriminant kvadriky ∆

> Delta:=det(K);

∆ := −972
Diskriminant je různý od nuly. Kvadrika je tedy regulární.

Hlavní minor kvadriky A44

> SubK:=submatrix(K,1..3,1..3);

SubK :=

⎡
⎣ 7 −2 0

−2 6 −2
0 −2 5

⎤
⎦

> A44:=det(SubK);

A44 := 162

Hlavní minor kvadriky A44 je různý od nuly. Kvadrika je tedy středová.

Charakteristická rovnice kvadriky

Nejprve vytvoříme jednotkovou matici E:

> E:=evalm(array(1..3,1..3,identity));

E :=

⎡
⎣ 1 0 00 1 0
0 0 1

⎤
⎦

Potom definujeme matici ChM příslušné homogenní soustavy, která vede k cha-
rakteristické rovnici ChR:

> ChM:=evalm(SubK-lambda*E);

ChM :=

⎡
⎣ 7− λ −2 0

−2 6− λ −2
0 −2 5− λ

⎤
⎦

> ChR:=det(ChM)=0;

ChR := 162 − 99λ+ 18λ2 − λ3 = 0

Kořeny charakteristické rovnice jsou vlastní čísla kvadriky:

> ResChR:=solve(ChR,lambda);

ResChR := 3, 6, 9

Pro snazší manipulaci můžeme vlastní čísla zapsat jako složky vektoru (uspo-
řádané trojice) λ :

> lambda:=[ResChR];

λ := [3, 6, 9]

Potom ke konkrétnímu vlastnímu číslu přistoupíme prostřednictvím odpoví-
dajícího indexu (pořadového čísla v uspořádané trojici):

> lambda[1]; lambda[2]; lambda[3];

3
6
9

Kanonický tvar rovnice kvadriky

Absolutní člen rovnice
∆

A44
:

> Delta/A44;

−6
> kr:=lambda[1]*x^2+lambda[2]*y^2+9*z^2=-Delta/A44;

kr := 3x2 + 6 y2 + 9 z2 = 6

Konečná podoba kanonického tvaru rovnice dané kvadriky:

> KanRovKv:=kr/(abs(Delta/A44));

KanRovKv :=
x2

2
+ y2 +

3 z2

2
= 1

Danou kvadrikou je trojosý elipsoid.

Délky poloos elipsoidu:

> a=sqrt(1/coeff(lhs(KanRovKv),x,2));

> b=sqrt(1/coeff(lhs(KanRovKv),y,2));

> c=sqrt(1/coeff(lhs(KanRovKv),z,2));

a =
√
2

b = 1

c =

√
6
3

Zobrazení elipsoidu v transformované soustavš souřadnic provedeme pomocí
funkce implicitplot

> implicitplot3d(KanRovKv,x=-2..2,y=-2..2,z=-2..2,grid=[30,30,30],

style=patchcontour,orientation=[40,55],axes=frame,tickmarks=[3,3,3],

color=COLOR(RGB,250/255,250/255,250/255),light=[90,-5,1,1,1]);

Vyšetření polohy kvadriky v původní soustavě souřadnic

Souřadnice středu kvadriky.
Uvažujme vzájemnou polohu přímky Primka a dané kvadriky RovKv:

> Primka:=[x=m+t*u,y=n+t*v,z=p+t*w];

> RovKv;

7x2 + 6 y2 + 5 z2 − 4x y − 4 y z − 22x+ 24 y + 2 z + 30 = 0

Dosazení parametrických rovnic přímky za x, y, a z do rovnice kvadriky vede
k následující rovnici s proměnnou t :

> RovKvPr:=simplify(eval(RovKv,Primka));

RovKvPr := 7m2 + 14m t u+ 7 t2 u2 + 6n2 + 12n t v + 6 t2 v2 + 5 p2 + 10 p t w

+5 t2w2 − 4m n − 4m t v − 4 t u n − 4 t2 u v − 4n p − 4n t w − 4 t v p

−4 t2 v w − 22m − 22 t u + 24n + 24 t v + 2 p+ 2 t w + 30 = 0

Rovnici můžeme zapsat ve tvaru At2 + B t+ C = 0, kde koeficienty A, B, C
mají následující tvar:

> A:=coeff(lhs(RovKvPr),t^2);

A := 7u2 + 6 v2 + 5w2 − 4u v − 4 v w

> B:=1/2*coeff(lhs(RovKvPr),t);

B := 7m u+ 6n v + 5 p w − 2m v − 2un − 2n w − 2 v p − 11u+ 12 v + w

> C:=coeff(lhs(RovKvPr),t,0);

C := 7m2 + 6n2 + 5 p2 − 4m n − 4n p − 22m+ 24n + 2 p+ 30

Středem kvadriky je bod S = [m, n, p], pro jehož souřadnice je koeficient B
roven nule, tj. B = 0, bez ohledu na souřadnice [u, v, w] směrového vektoru
přímky.

> B1:=collect(B,[u,v,w]);

B1 := (7m − 11− 2n)u+ (6n − 2m+ 12 − 2 p) v + (−2n+ 5 p + 1)w
Souřadnice středu kvadriky tak určíme řešením následující homogenní sou-
stavy rovnic s neznámými m, n, p :

> RStr:=coeffs(B1,[u,v,w]);

RStr := 6n − 2m+ 12− 2 p, −2n + 5 p + 1, 7m − 11 − 2n
> RStr_res:=solve({RStr},{m,n,p});

RStr res := {m = 1, p = −1, n = −2}
Střed kvadriky:

> S:=eval([m,n,p],RStr_res);

S := [1, −2, −1]

Hlavní směry kvadriky

Řešíme příslušné homogenní soustavy rovnic, postupně pro všechny tři vlastní
čísla.

1) λ1 = 3

> MatHlSm1:=evalm(SubK-lambda[1]*E);

MatHlSm1 :=

⎡
⎣ 4 −2 0

−2 3 −2
0 −2 2

⎤
⎦

> RovHlSm1:=geneqns(MatHlSm1,[u,v,w]);

RovHlSm1 := {−2u+ 3 v − 2w = 0, −2 v + 2w = 0, 4u − 2 v = 0}
> HlSm1:=solve(RovHlSm1,{u,v,w});

HlSm1 := {v = 2u, w = 2u, u = u}
> u1:=eval([u,v,w],HlSm1);

u1 := [u, 2u, 2u]

Hlavní směr u1 :

> u1:=eval(u1,u=1);

u1 := [1, 2, 2]

2) λ2 = 6

> MatHlSm2:=evalm(SubK-lambda[2]*E);

MatHlSm2 :=

⎡
⎣ 1 −2 0

−2 0 −2
0 −2 −1

⎤
⎦

> RovHlSm2:=geneqns(MatHlSm2,[u,v,w]);

RovHlSm2 := {u − 2 v = 0, −2 v − w = 0, −2u − 2w = 0}
> HlSm2:=solve(RovHlSm2,{u,v,w});

HlSm2 := {u = 2 v, w = −2 v, v = v}
> u2:=eval([u,v,w],HlSm2);

u2 := [2 v, v, −2 v]
Hlavní směr u2 :

> u2:=eval(u2,v=1);

u2 := [2, 1, −2]

3) λ3 = 9

> MatHlSm3:=evalm(SubK-lambda[3]*E);

MatHlSm3 :=

⎡
⎣ −2 −2 0

−2 −3 −2
0 −2 −4

⎤
⎦

> RovHlSm3:=geneqns(MatHlSm3,[u,v,w]);

RovHlSm3 := {−2u − 2 v = 0, −2u − 3 v − 2w = 0, −2 v − 4w = 0}
> HlSm3:=solve(RovHlSm3,{u,v,w});

HlSm3 := {u = 2w, v = −2w, w = w}
> u3:=eval([u,v,w],HlSm3);

u3 := [2w, −2w, w]

Hlavní směr u3 :

> u3:=eval(u3,w=1);

u3 := [2, −2, 1]

Hlavní roviny

Odvodíme obecnou rovnici průměrové roviny sdružené se směrem [u, v, w]

> U:=Vector[row]([u,v,w,0]);

U := [u, v, w, 0]

> PrumerR:=collect(expand(evalm(U&*K&*Transpose(X))),[x,y,z])=0;

PrumerR := (7u − 2 v)x + (−2u+ 6 v − 2w) y + (−2 v + 5w) z − 11u+ 12 v + w = 0

Postupným dosazením hlavních směrů dostaneme obecné rovnice příslušných
hlavních rovin:

> HlR1:=eval(PrumerR,[u=u1[1],v=u1[2],w=u1[3]]);

> HlR2:=eval(PrumerR,[u=u2[1],v=u2[2],w=u2[3]]);

> HlR3:=eval(PrumerR,[u=u3[1],v=u3[2],w=u3[3]]);

HlR1 := 3x+ 6 y + 6 z + 15 = 0

HlR2 := 12x+ 6 y − 12 z − 12 = 0
HlR3 := 18x − 18 y + 9 z − 45 = 0

Zobrazení kvadriky v původní poloze spolu s jejími osami a hlavními
rovinami

Můžeme si definovat barvu(y) pro obarvení grafu(ů):

> col1:=COLOR(RGB,250/255,250/255,250/255):

Jednotlivé grafy uložíme do proměnných:

1) Kvadrika:

> Kvg:=implicitplot3d(RovKv,x=-2..3,y=-4..0,z=-3..2,axes=frame,

color=col1,style=patchnogrid,grid=[40,40,40],light=[60,20,1,1,1],

tickmarks=[3,3,3],orientation=[52,63],scaling=constrained):

2) Osy elipsoidu:

> o1g:=plot3d(evalm(S+t*u1),t=-2..2,j=-1..1,thickness=3):

> o2g:=plot3d(evalm(S+t*u2),t=-2..2,j=-1..1,thickness=3):

> o3g:=plot3d(evalm(S+t*u3),t=-2..2,j=-1..1,thickness=3):

3) Hlavní roviny:

> HlR1g:=plot3d(solve(HlR1,z),x=-2..3,y=-4..0,color=grey,

style=patchnogrid,contours=60):

> HlR2g:=plot3d(solve(HlR2,z),x=-2..3,y=-4..0,color=red,

style=patchnogrid,contours=60):

> HlR3g:=plot3d(solve(HlR3,z),x=-2..3,y=-4..0,color=green,

style=patchnogrid,contours=60):

K zobrazení více grafů v jedné soustavě použijeme příkaz plots[display]:

> display(Kvg,o1g,o2g,o3g,HlR1g,HlR2g,HlR3g,axes=frame,

scaling=constrained,orientation=[143,75],view=[-2..3,-4..0,-3..2]);

Závěr

V této publikaci jsme se seznámili se základy teorie kvadrik, která navazuje
na teorii kuželoseček. Po úvodní teoretické části, která se zabývá obecnými
vlastnostmi kvadratických ploch, jsou ve druhé části studovány vlastnosti jed-
notlivých kvadrik se zřetelem k jejich použití v praxi.
Svými jedinečnými vlastnostmi jsou kvadriky nedílnou součástí našeho života.

